Kom godt i gang med Maple 12

(Worksheet mode)

Indtastning af taludtryk - simple beregninger

Anbring cursoren ved en *input-linje*, dvs. en linje med symbolet > yderst til venstre. Er dette symbol ikke til stede, så skaffer du et ved at trykke på knappen (i værktøjslinjen)



Skriv her 2+2, og tryk Enter:

> $2+2$

$$4 \tag{1}$$

Du kan indtaste i to modes: **1D- og 2D-Math**. Du skifter mellem disse med **F5**-knappen. Når du trykker på F5-knappen, skifter cursoren udseende fra en skrå streg (/) til en lodret streg (|), så du kan altid på cursoren se, hvilken mode, du befinder dig i. I 1D-Math mode ser ovenstående regnestykke således ud:

```
> 2+2;
```

$$4 \tag{2}$$

I 2D-Math mode behøver du ikke at afslutte indtastninger med semi-kolon. Bruges i stedet kolon som afslutning, vil udregningen blive foretaget, men resultatet vises ikke:

> $2+3\cdot 5:$

At der er sket en udregning af udtrykket kan du se ved at bede Maple vise det sidst udregnede resultat. Som reference til det sidst udregnede resultat benytter Maple et %-tegn. Dette virker uanset om resultatet er vist eller ej:

> %

$$17 \tag{3}$$

Advarsel: Vær varsom med brugen af %, da det sidst udregnede resultat ikke nødvendigvis stammer fra linjen ovenover. Du kan nemlig til enhver gå op i en input-linje, eventuelt lave ændringer, og så snart du taster Enter, så sker der en genberegning af udtrykket, og % vil få en ny værdi. Prøv!

Arbejdsområde 1

I dette område er det meningen, at du skal afprøve de ovenstående ting - herunder undersøge, hvordan % virker. Hvad sker der, hvis du taster %%% ? Har du ikke plads nok, kan du altid tilføje mere plads ved at trykke på [> - knappen.

>

>
>
>
>

Det er vigtigt, at du bliver fortrolig med Maples formeleditor. Skal du fx indtaste et udtryk som $\frac{17-2}{5}$, kan du naturligvis taste ind ved at sætte en parentes om tælleren, altså som (17-2)/5

> $\frac{(17-2)}{5}$

$$3 \tag{4}$$

Det ser bare ikke så kønt ud med denne overflødige parentes i tælleren. Men Maple kan gøre det bedre:

• Indtast først tælleren $17-2$.

• Marker hele tælleren ved at trække med musen - eller alternativt: Hold Shift-tasten nede, og tryk på pil-venste et par gange til hele udtrykket er markeret.

• Med hele tælleren markeret taster du / for division. Straks vil en pæn brøk komme frem med cursoren stående i nævneren.

• Indtast nævneren, og tast Enter.

> $\frac{17-2}{5}$

$$3 \tag{5}$$

Arbejdsområde 2

Indtast ovenstående brøk og check, at alt virker som det skal.

Indtast også udtrykket $\frac{29+17}{25}+\frac{3}{7+4}$. Der opstår sikkert et problem, når du skal indtaste + , idet din cursor er placeret i nævneren i den første brøk. Her gør pil-højre tasten underværker.

>
>
>
>
>

Skal du indtaste mere komplicerede udtryk, fx potenser, kvadratrødder, n-te rødder mv., så klik på pilen i Expression-knappen (i skærmens højre side):

Dette vil åbne Expression-paletten, hor du finder skabeloner til at lette indtastningen. Skal du fx skrive potens 2^4, så klikker du på knappen



dette vil give dig denne skabelon i indtastningslinjen:



du indtaster et 2-tal, og trykker herefter på TAB-knappen på tastaturet. Eksponenten *b* vil så blive markeret med blåt, og du kan indtaste 4-tallet. Sådan virker i princippet alle skabeloner med mere end én pladsholder.

> 2^4

$$16 \tag{6}$$

Du finder også en brøk-skabelon i Expression paletten. Tryk på denne, og du får en skabelon som denne $\frac{a}{b}$ med tælleren markeret. Skriv tælleren, og tryk på tabulator knappen for at hoppe til nævneren. Prøv!

Arbejdsområde 3

Eksperimenter med potens- og brøk-skabelonerne, og indtast herefter udtrykkene

- $-3.17 + \dfrac{2.53^2 - \sqrt{5.25}}{2.46}$
- $(1-2\sqrt{3})^2 + \left(\sqrt{3}+3\right)^2$

>
>
>
>
>

Som du måske har bemærket, så returnerer Maple altid - om muligt - det eksakte resultat. Dvs., at hvis der forekommer et decimaltal i udtrykket, så returneres resultatet som et decimaltal, ellers returneres det eksakte resultat. Du kan tvinge Maple til at aflevere et tilnærmet resultat ved at benytte kommandoen **evalf**. Det fungerer således:

> $evalf\left(\dfrac{1+\sqrt{5}}{2}\right)$

$$1.618033988 \tag{7}$$

- og vil du have resultatet med fx 4 cifre, så skriver du

> $evalf[4]\left(\dfrac{1+\sqrt{5}}{2}\right)$

$$1.618 \tag{8}$$

Arbejdsområde 4

Udregn nedenstående udtryk, og afgør, om de er hele tal.
OBS: π og e (den naturlige logaritmes grundtal) skal skrives vha. **Common Symbols**-paletten

- $sin\left(2017\cdot\sqrt[5]{2}\right)$

- $e^{\pi\cdot\sqrt{163}} - 640320^3$

>
>
>
>
>

Arbejde med tekst i et Maple dokument

Det dokument, du sidder med, er skrevet i Maple. Beregningsregioner indledes med >, hvorimod tekstregioner er er uden >. Du indsætter en tekstregion med knappen

At skrive tekst i Maple er som at skrive i et almindeligt tekstbehandlingsprogram, bortset fra, at du her har en formeleditor indbygget - og det er nøjagtig den samme som den, du bruger i beregningsregioner:

For at indsætte matematik i en tekstregion taster du F5, hvorved cursoren bliver skråtstillet. Du indtaster det matematiske udtryk, og afslutter med F5. Herved er du tilbage i tekstmode igen.

TIP: Du kan ændre en tom beregningsregion til en tekstregion ved at taste **Ctrl+t**. En tom tekstregion ændres til en beregningsregion ved at taste **Ctrl+m**.

Arbejdsområde 5

Skriv denne tekst i Maple:

Tredjegradsligningen $x^3 + c\cdot x = d$ kan løses ved at skære en parabel med ligningen $y = \sqrt{c}\cdot y$ og en cirkel med centrum i $\left(\frac{d}{2\,c}, 0\right)$ og radius $\frac{d}{2\,c}$.

Regning med symbolske udtryk

Maple kan naturligvis også regne med symbolske udtryk - dvs. udtryk, hvor der indgår en række symboler, der ikke er tildelt nogen værdi. Fx

> $x + 2\,x + 3\,x + 4\,y + 5\,y + 6\,y$

$$6\,x + 15\,y \tag{9}$$

her samles automatisk x'erne og y'erne. Der ganges også automatisk ind i parenteser med tal, men ikke med symboler:

> $2(3x+2)$

$$6x+4 \qquad (10)$$

> $x\cdot(3x+4)$

$$x\,(3x+4) \qquad (11)$$

I **(11)** er det meget vigtigt at du skriver et gange-tegn foran parentesen. I modsat fald vil Maple opfatte udtrykket som funktion med navnet *x*, der skal udregnes på $3x+4$.

Du kan tvinge Maple til at gange parenteser ud (- altså at omskrive til ledform) ved at bruge kommandoen *expand*:

> $expand(x\cdot(3x+4))$

$$3x^2+4x \qquad (12)$$

Brøken $\dfrac{a^2-b^2}{a+b}$ kan forkortes $a-b$. Det ses nemmest ved at omskrive tælleren til $(a+b)\cdot(a-b)$. Men kan vi få Maple til det? Der sker ingen automatisk forkortning (det ville ske, hvis der kun indgik tal):

> $\dfrac{a^2-b^2}{a+b}$

$$\frac{a^2-b^2}{a+b} \qquad (13)$$

Expand giver heller ikke det ønskede, men naturligvis det forventede, idet brøken nu er skrevet som en differens af to led (altså bragt på ledform).

> $expand\left(\dfrac{a^2-b^2}{a+b}\right)$

$$\frac{a^2}{a+b}-\frac{b^2}{a+b} \qquad (14)$$

Derimod er kommandoen *normal* skabt til opgaven.

> $normal\left(\dfrac{a^2-b^2}{a+b}\right)$

$$a-b \qquad (15)$$

Kommandoen *simplify* kan også bruges her til at sætte på fælles brøkstreg. Denne forenkler - om muligt - udtrykket mest muligt. Dette kan også indbefatte at gange parenteser ud.

Den sidste kommando til manipulation af symbolske udtryk, vi vil se på her, er *factor*, der faktoriserer et symbolsk udtryk:

> $factor(a^2-b^2)$

$$(a-b)\,(a+b) \qquad (16)$$

> $factor(x^2 + 4x + 3)$

$$(x+3)(x+1) \quad (17)$$

Arbejdsområde 6

- Omskriv med passende Maple kommandoer kvadratet på en to-leddet størrelse $(a+b)^2$ til formen $a^2 + b^2 + 2\,a \cdot b$

- Addition af talbrøker , fx $\frac{2}{3} + \frac{5}{7}$, vil altid ske automatisk i Maple. Prøv! Derimod er der ingen automatik, hvis du forsøger dig med fx $\frac{1}{a} + \frac{1}{b}$. Hvordan skal man så få dem på fælles brøkstreg?

>
>
>

Variabler og formler

Du kan gemme et udtryk i en variabel ved at benytte **tildelingssymbolet :=**, der er et kolon efterfulgt af et lighedstegn. Fx vil tildelingen

> $abc := 14\,x^3 + \frac{7}{x \cdot y^3}$

$$abc := 14\,x^3 + \frac{7}{x\,y^3} \quad (18)$$

bevirke, at at udtrykket $14\,x^3 + \frac{7}{x \cdot y^3}$ gemmes i variablen med navnet *abc*. Herefter kan *abc* benyttes i stedet for det oprindelige udtryk. Fx

> $\frac{abc}{7}$

$$2\,x^3 + \frac{1}{x\,y^3} \quad (19)$$

Læg mærke til, at det er det symbolske udtryk, der gemmes i variablen *abc*. Hvis *x* og *y* får tildelelt talværdier, så vil *abc* beregnes til en talværdi. Tildel *x* værdien 2, og *y* værdien 3. Dette sker også med tildelingssymbolet :=

> $x := 2;\, y := 3$

$$x := 2$$
$$y := 3 \quad (20)$$

hvor kolon (:) efter tildelingerne betyder, at vi ikke ønsker at se output, der jo blot ville være i stil med $x = 2$ og $y = 3$. Beregner du nu *abc*, får du et tal:

> *abc*

(21)

$$\frac{6055}{54} \tag{21}$$

Når du tildeler værdi med $x := 2$ og $y := 3$, så har x og y disse værdier indtil du giver dem nye værdier ved en ny tildeling. Da specielt x og y ofte benyttes i beregninger, er det ikke smart, at de har faste værdier. Du kan rense x og y ved at skrive:

> $x := 'x'; y := 'y'$

$$x := x$$
$$y := y \tag{22}$$

Check, at x og y er rensede, og at *abc* atter beregner til formlen i **(18)**.

> $x; y; abc$

$$x$$
$$y$$
$$14\,x^3 + \frac{7}{x\,y^3} \tag{23}$$

Du kan lave en tildeling af værdier til x og y, der kun er aktiv så længe beregningen af *abc* står på. Dette kaldes en lokal tildeling. Hertil skal du benytte *eval* - kommandoen:

> $eval(abc, \{x = 2, y = 3\})$

$$\frac{6055}{54} \tag{24}$$

Arbejdsområde 7

- Prøv at tildele nye værdier til x og y i linje **(20)** (husk at taste Enter efter tildelingen, så linjen bliver udført!). Beregn så *abc* ved at gå til linjen med *abc*, og taste Enter.

- Hvis du ikke vil se output fra tildelingen af værdier til x og y, skal du i stedet skrive $x := 2 : y := 3$: Prøv!

- Rumfanget af en tønde med højde h, radius R og endefladeradius r beregnes med formlen:

$$V = \frac{\pi \cdot h}{15}\left(8\,R^2 + 4\,r \cdot R + \frac{3}{4}\,r^2\right)$$

Find rumfanget af en tønde, når $h = 11$, $r = 4$ og $R = 5$. Prøv med begge former for tildeling.

>
>
>
>
>

Undertiden får du behov for at få Maple til at glemme alle tildelinger, du har lavet. Dette klarer du nemt ved at fyre *restart* kommandoen af (der kommer ingen output fra denne kommando):

> *restart*

> abc, r, R

$$abc, r, R \tag{25}$$

Som det ses, har Maple glemt alt om tildelingen af værdier til variablerne *abc*, *r* og *R*.

Du kan føje kommentarer til en kommando linje ved at placere # foran kommentaren, fx

> *restart #Maple glemmer alt.*

Ligningsløsning

Ligningsløsning går for det meste let og smertefrit i Maple, men dog langt fra altid! Du skal nu løse ligningen $x = \frac{1}{x-1}$. Det er ofte en rigtig god ide at navngive ligningen (dvs. gemme den i en variabel):

> $Eq := x = \frac{1}{x-1}$

$$Eq := x = \frac{1}{x-1} \tag{26}$$

Eksakt løsning af denne ligning sker med $solve(Eq, x)$, hvor x'et refererer til, at ligningen løses med hensyn til x (hvis der, som her, kun er en variabel i ligningen, kan ', x ' udelades):

> $solve(Eq, x)$

$$\frac{1}{2} + \frac{1}{2}\sqrt{5}, \frac{1}{2} - \frac{1}{2}\sqrt{5} \tag{27}$$

Der er to løsninger, adskilt ved et komma. Denne opskrivning hedder i Maple terminologi en udtryksfølge (*expression sequence*). Også her kan det være nyttigt at gemme løsningerne i en variabel. Fx

> $sol := solve(Eq, x)$

$$sol := \frac{1}{2} + \frac{1}{2}\sqrt{5}, \frac{1}{2} - \frac{1}{2}\sqrt{5} \tag{28}$$

så kan du efterfølgende få fat i løsningerne ved at skrive hhv. $sol[1]$ og $sol[2]$:

> $sol[1]$

$$\frac{1}{2} + \frac{1}{2}\sqrt{5} \tag{29}$$

Alternativt kan du skrive sol_1 (der kommer til at se sådan ud sol_1):

> sol_1

$$\frac{1}{2} + \frac{1}{2}\sqrt{5} \tag{30}$$

Løsning af flere ligninger med flere ubekendte sker tilsvarende, blot skal du samle ligningerne i krøllede parenteser {}, og tilsvarende for de variabler, du løser med hensyn til. Et eksempel:

Løs ligningssystemet:

$$\begin{cases} x^2 + 2x + y^2 - 4y = 3 \\ y = 4x \end{cases}$$

> $Eq1 := x^2 + 2x + y^2 - 4y = 3$

$$Eq1 := x^2 + 2x + y^2 - 4y = 3 \tag{31}$$

> $Eq2 := y = 4x$

$$Eq2 := y = 4x \tag{32}$$

> $solve(\{Eq1, Eq2\}, \{x, y\})$

$$\left\{x = -\frac{3}{17}, y = -\frac{12}{17}\right\}, \{x = 1, y = 4\} \tag{33}$$

Geometrisk svarer opgaven til at finde skæringspunkterne mellem en cirkel og en ret linje.
Skæringspunkterne er altså

$$\left(-\frac{3}{17}, -\frac{12}{17}\right), \ (1, 4)$$

Arbejdsområde 8

- Løs den generelle andegradsligning $a \cdot x^2 + b \cdot x + c = 0$ med hensyn til x.

- Løs ligningssystemet (med hensyn til x og y):

$$\begin{cases} x + y = a \\ -y + 4x = b \end{cases}$$

- Tildel a og b nogle værdier, og løs ovenstående ligningssystem igen.

- Isoler R, r og h i ligningen

$$V = \frac{\pi \cdot h}{15}\left(8R^2 + 4r \cdot R + 3r^2\right)$$

>
>
>
>
>

Uligheder

Uligheder løses ligesom ligninger - blot skal du anvende et ulighedstegn: <, >, ≤ eller ≥ - hvor de første to sidder på tastaturet, mens de to sidste findes i Common-symbols paletten.

Lad os se på et eksempel:

$> \ solve\left(-\frac{1}{2}x^2 - 6\,x - 16 \le 2\,x + 14, x\right)$

$$RealRange(-\infty, -10), RealRange(-6, \infty) \qquad \textbf{(34)}$$

Løsningen skal her opfattes som $]-\infty, -10] \cup [-6, \infty[$.

$> \ solve\left(-\frac{1}{2}x^2 - 6\,x - 16 < 2\,x + 14, x\right)$

$$RealRange(-\infty, Open(-10)), RealRange(Open(-6), \infty) \qquad \textbf{(35)}$$

Her skal løsningen tilsvarende opfattes som $]-\infty, -10[\cup]-6, \infty[$.

Labels

Du har sikkert bemærket de numre, der står ud for hvert resultat i højre side af skærmen. Disse numre kaldes **Labels**, og dem kan du bruge, når du vil referere til tidligere resultater. Disse resultater kan være tal, tildelinger, løsninger mv.

$> \ 2 + 3$ *#resultatet bliver et tal*

$$5 \qquad \textbf{(36)}$$

$> \ a := x + y + z$ *#resultatet bliver en tildeling*

$$a := x + y + z \qquad \textbf{(37)}$$

$> \ solve(x^2 + 2\,x - 2 = 0, x)$ *#resultatet bliver en udtryksfølge*

$$-1 + \sqrt{3}, -1 - \sqrt{3} \qquad \textbf{(38)}$$

Hvis du vil referere til et af disse resultater, taster du **Ctrl+l** (dvs., du holder Ctrl-tasten nede, og trykker på l - l som i label). Dette bevirker, at en lille boks popper op:



I Label Value feltet skriver du nummeret - uden parenteser! Labels opdateres automatisk, hvis du putter en ny udregning ind i en eksisterende følge.

$>$ **(36)**

$$5 \qquad \textbf{(39)}$$

$>$ **(37)**

(40)

$$x + y + z \qquad \textbf{(40)}$$

> **(38)**

$$-1 + \sqrt{3}, -1 - \sqrt{3} \qquad \textbf{(41)}$$

Labels kan indgå lige som variabler i udtryk (under forudsætning af, at det giver mening):

> $\textbf{(37)}^2 + \textbf{(36)}^2$

$$(x + y + z)^2 + 25 \qquad \textbf{(42)}$$

Hvis resultatet er en udtryksfølge, kan du udtrække de enkelte elementer således:

> $\textbf{(38)}_2$

$$-1 - \sqrt{3} \qquad \textbf{(43)}$$

Arbejdsområde 9

Ved løsning af et system af 2 ligninger med 2 ubekendte bliver resultatet

> $solve(\{x^2 + 2x + y^2 - 4y = 3, y = 4x\}, \{x, y\})$

$$\left\{x = -\frac{3}{17}, y = -\frac{12}{17}\right\}, \{x = 1, y = 4\} \qquad \textbf{(9.1)}$$

Prøv at udtrække $y = -\frac{12}{17}$ ved at benytte label referencen **(9.1)** og de kantede parenteser []. Læg mærke til, at label referencen ser lidt anderledes ud her end ovenfor, hvor det blot var et enkelt ciffer. **(9.1)** refererer til, at du er i underafsnit nr. 7 og resultatet er det første i denne sektion.

>
>
>
>
>

Funktioner

> *restart*

Funktioner i Maple skal defineres efter syntaksen

$$f := x \rightarrow udtryk\ i\ x$$

Du kan benytte skabelonen $f := a \rightarrow y$ fra Expression-paletten når du skal definere funktioner, men da du skkert ofte kommer ud for at skulle definere en funktion, kan du lige så godt lære genvejen med det samme: Pilen laves ved at taste -> (altså en bindestreg efterfulgt af et 'større end' tegn).

Definer en funktion ved:

> $f := x \rightarrow x^2 - 2\,x + 5$

$$f := x \rightarrow x^2 - 2\,x + 5 \qquad \textbf{(44)}$$

Herefter er funktionen klar til brug. Du kan fx udregne nogle funtionsværdier:

> $f(2);\ f\left(\sqrt{5} + 3\right);\ f(a + b)$

$$5$$
$$\left(\sqrt{5} + 3\right)^2 - 2\sqrt{5} - 1$$
$$(a + b)^2 - 2\,a - 2\,b + 5 \qquad \textbf{(45)}$$

Der er et væld af funktioner indbgget i Maple. Du kan få en oversigt i hjælpesiderne, som du får adgang til via menuen **Help > Maple Help**. Prøv her at søge efter *function*, og find oversigten *function,index* i søgeresultaterne.

Hvis du ved, hvad du skal have fat i fra hjælpesiderne, kan du med ?-kommandoen få den ønskede hjælpeside frem straks:

> ?*index,function*

Hvis du prøver at definere en funktion ved $f(x) := $ *udtryk i x*, brokker Maple sig, og spørger, om du er klar over, hvad du er i færd med.

Ofte udelader man i matematik at skrive gangetegn mellem en variabel og en parentes, fx skriver man $a(b + c)$, selvom man faktisk mener $a \cdot (b + c)$. Den går ikke i Maple!

> $a \cdot (b + c)$

$$a\ (b + c) \qquad \textbf{(46)}$$

> $a(b + c)$

$$a(b + c) \qquad \textbf{(47)}$$

Læg mærke til forskellen i output. Maple skriver ikke gangetegnet i **(46)**, men sætter i stedet et lille mellemrum ind. I **(47)** opfattes a som navnet på en funktion, der skal anvendes på summen $a + b$. Gør det derfor til en vane, at skrive alle gangetegn - undtagen dem, der står mellem et tal og en variabel, som fx $2\,x$.

I Maple skal du bruge funktionsparenteser. I matematikbøger skriver man ofte fx ln x i stedet for $\ln(x)$, og tilsvarende sin x i stedet for $\sin(x)$. Det går ikke i Maple!

Arbejdsområde 10

En parabel går gennem punkterne $(1, 2)$, $(2, 1)$ og $(3, 3)$. Find forskriften.
Vejledning: Start med i Maple at definere det generelle andengradspolynomium ved $p(x) = ax^2 + bx + c$, og opskriv vha. forskriften de 3 ligninger med 3 ubekendte, de givne punkter giver anledning til. Løs dette ligningssystem.

> *restart*
>

>
>
>

Graftegning

Det er meget simpelt at tegne funktionsgrafer i Maple. En enkelt kommando - *plot* - klarer det meste. Definer en funktion ved

> $f := x \rightarrow x^2 - 6$

$$f := x \rightarrow x^2 - 6 \tag{48}$$

Grafen kan tegnes i et standardvinduet, hvor *x*-intervallet er $[-10, 10]$, og *y*-intervallet er tilpasset *x*-intervallet (Autoskaleret).

> $plot(f(x))$



Hvis du vil ændre *x*-intervallet, angiver du dette som en parameter til plotkommandoen. Fx skal *x*-

intervallet $[-5, 5]$ angives som $x=-5..5$, hvor de to prikker laves med to tryk på punktum-tasten.

> $plot(f(x), x=-5..5)$



Tilsvarende kan du kontrollere y-intervallet

> $plot(f(x), x=-5..5, y=-10..10)$



Hvis du vil tegne en graf mere i samme koordinatsystem, så skal pakke de to funktioner sammen med [] eller {}. Benyt {}, hvis den rækkefølge, hvori graferne tegnes, er ligegyldig:

> $g := x \rightarrow -\frac{3}{2}x + 2$

$$g := x \rightarrow -\frac{3}{2}\,x + 2 \qquad \textbf{(49)}$$

> $plot(\{f(x), g(x)\}, x = -5..5, y = -10..10)$



Arbejdsområde 11

Klik i graffeltet ovenfor. I det aktive grafområde bliver markøren til et sigtekorn. Før sigtekornet hen til et af skæringspunkterne, og du kan aflæse nogle tilnærmede værdier øverst i Maplevinduet:

I dette vindue kan du lave en række ændringer i koordinatsystemet. Afprøv nogle af disse muligheder.

Hvis du med sigtekornet klikker på en af graferne, markeres denne. Ved et højre-klik kan du så åbne for kontekst-menuen, hvor du finder en række indstillinger for den enkelte graf. Skift fx farve og tykkelse.

Klik på **Drawing**-knappen, så skifter menuen til denne

Her kan du fx tilføje tekst til dine grafer. Prøv.

Alle ændringer, du har lavet ovenfor, er midlertidige. Hvis du gentegner plottet (gå til plot-kommandoen ovenfor, og tast Enter), så forvinder alle dine ændringer. Skal ændringerne være permanente, så skal de tilføjes selve plot-kommandoen. Hvis farverne skal være rød og blå, ændres plot-kommandoen til (fjern : i slutningen af linjen og taste Enter, hvis du vil se plottet)

> $plot(\{f(x), g(x)\}, x = -5..5, y = -10..10, color = [blue, green])$:

Er farverækkefølgen den forventede? Ellers lav de fornødne ændringer.

Vil du lave stiplede linjer mm. er kommandoen denne:

> $plot(\{f(x), g(x)\}, x = -5..5, y = -10..10, color = [blue, green], linestyle = [dash, solid])$:

Alle detaljer finder du her
> ?*plot,options*

Eksempel - Overlevelse

Nedenstående funktion er et eksempel på en såkaldt overlevelseskurve

> *restart*
> $L := x \rightarrow 100 \cdot e^{0.0005(1-x) - 0.000867(1.0914^x - 1.0914)}$

$$L := x \rightarrow 100\, e^{0.0005\,(1-x) + (-1)\cdot 0.000867\,(1.0914^x - 1.0914)} \quad \textbf{(50)}$$

Fx er

> $L(65)$

$$75.10300090 \quad \textbf{(51)}$$

Dette skal tolkes således, at ved en alder på 65 år er 75% i live, altså 25% er døde.

Grafen ser således ud (med en masse pynt):

> $plot(L(x), x = 0..100, axes = boxed, title = \text{'Overlevelseskurve'}, titlefont = [TIMES, ITALICS, 18], labels = [alder, pct\ overlevende], labeldirections = [horizontal, vertical])$



Arbejdsområde 12

1. Hvor stor en procentdel er i live efter 80 år?

2. Hvornår er halvdelen faldet fra?

3. I hvilket år er dødeligheden størst?
   Vejledning: Se på funktionen $d(x) = L(x+1) - L(x)$, og indtegn denne i et nyt koordinatsystem.

>
>
>
>
>

Vi vil nu prøve at undersøge betydningen af konstanten 0.005 i forskriften for *L*. Derfor parametriserer vi *L* med hensyn til denne konstant, og kalder den nye funktion *L1*. Denne er så en funktion af to variabler:

$> L1 := (x, A) \rightarrow 100 \cdot e^{A \cdot (1-x) - 0.000867 (1.0914^x - 1.0914)}$

$$L1 := (x, A) \rightarrow 100 \, e^{A (1-x) + (-1) \cdot 0.000867 (1.0914^x - 1.0914)} \quad \textbf{(52)}$$

$L1(x, 0.0005)$ er således identisk med funktionen $L(x)$. Vi tegner $L1$ for forskellige værdier af A:

$> plot([L1(x, 0.01), L1(x, 0.005), L1(x, 0.0005)], x = 0..100, color = [red, green, blue])$



Grafen for $L1$ bliver en 3D-flade, og de tre kurver, vi har tegnet ovenfor, kan opfattes som snit i denne flade. Lokalisér disse snit i fladen nedenfor.

$> plot3d(L1(x, A), x = 0..100, A = 0..0.05, axes = Framed)$



Du kan rotere fladen ved at trække i den.

Differentialregning

Maple kan naturligvis også differentiere symbolsk. Til udregning af differentialkvotienter benyttes skabelonen $\frac{d}{dx} f$ fra Expression-paletten. Benyt tabulatoren til at hoppe fra pladsholder til pladsholder ved udfyldningen:

> $\frac{d}{dx} x^2$

$$2x \tag{53}$$

> $\frac{d}{dx} \left(2x^3 + e^{-2x}\right)$

$$6x^2 - 2e^{-2x} \tag{54}$$

Ofte har du allerede defineret en funktion som fx

> $f := x \rightarrow e^x - 2x - 2$

(55)

$$f := x \rightarrow e^{x} - 2\,x - 2 \qquad \mathbf{(55)}$$

og skal bruge den afledede funktion. Denne har du direkte adgang til via *f'*:

> $f'(x)$

$$e^{x} - 2 \qquad \mathbf{(56)}$$

Du bruge den afledede funktion *f'* som enhver anden funktion - herunder fx bestemme bestemme nulpunkter

> $solve(\,f'(x) = 0, x)$

$$\ln(2) \qquad \mathbf{(57)}$$

Meget illustrativt kan det også være at tegne grafen for *f'* sammen med grafen fof *f*. Læg mærke til, at der, hvor *f'* er negativ, er *f* aftagende, og der, hvor *f'* er positiv, er grafen for *f* voksende. Desuden har *f* vandret tangent i de punkter, hvor *f'* er lig med 0.

> $plot(\,[\,f(x), f'(x)\,], x = -3\,..3, linestyle = [solid, dash])$



Du kan nemt finde en tangent til grafen for f i punktet med 1. koordinaten $x = 2$ vha tangentligningen:

> $t := x \rightarrow f'(2) \cdot (x-2) + f(2)$

$$t := x \rightarrow \left(\frac{d}{dx} f(x) \bigg|_{x=2} \right) (x-2) + f(2) \tag{58}$$

> $t(x)$

$$\left(e^2 - 2\right) (x-2) + e^2 - 6 \tag{59}$$

> $plot([f(x), t(x)], x = -3..3, y = -5..15)$



Arbejdsområde 13

Find vha. differentialregning maksimum og minimum for funktionen

$$f(x) = \frac{x}{x^2 + 3}$$

og tegn grafen for f og for den afledede funktion i samme koordinatsystem.

Gå dernæst til Tools > Tutors > Calculus - Single Variable > Curve Analysis, og skift

standardfunktionen ud med funktionen ovenfor. Udforsk dette værktøj.
>
>
>
>
>